Matrix Exponential via Clifford Algebras

Rafał ABŁAMOWICZ

* *Department of Mathematics and Physics, Gannon University, Erie, PA 16541*
E-mail: ablamowicz@gannon.edu

Received March 17, 1998; Accepted May 15, 1998

Expanded version of a talk presented at the Special Session on 'Octonions and Clifford Algebras', 1997 Spring Western Sectional 921st Meeting of the American Mathematical Society, Oregon State University, Corvallis, OR, 19–20 April 1997.

Abstract

We use isomorphism φ between matrix algebras and simple orthogonal Clifford algebras $C\ell(Q)$ to compute matrix exponential e^A of a real, complex, and quaternionic matrix A. The isomorphic image $p = \varphi(A)$ in $C\ell(Q)$, where the quadratic form Q has a suitable signature (p, q), is exponentiated modulo a minimal polynomial of p using Clifford exponential. Elements of $C\ell(Q)$ are treated as symbolic multivariate polynomials in Grassmann monomials. Computations in $C\ell(Q)$ are performed with a Maple package 'CLIFFORD'. Three examples of matrix exponentiation are given.

1 Introduction

Exponentiation of a numeric $n\times n$ matrix A is needed when solving a system of differential equations $\mathbf{x}' = A\mathbf{x}$, $\mathbf{x}(0) = \mathbf{x}_0$, in order to represent its solution in a form $\mathrm{e}^{At}\mathbf{x}_0$. It is well known that the exponential form of the solution remains valid when A is not diagonalizable, provided the following definition of e^A is adopted:

$$\mathrm{e}^A = \sum_{k=0}^{\infty} \frac{A^k}{k!}, \qquad \text{where} \quad A^0 = I. \tag{1}$$

Equation (1) means that the sequence of partial sums $S_n \equiv \sum_{k=0}^{n} A^k/k! \to \mathrm{e}^A$ entrywise. Equivalently, (1) implies that $\|S_n - \mathrm{e}^A\|_1 \to 0$ where $\|A\|_1$ denotes matrix 1-norm defined as the maximum of $\{\|A_j\|_1,\ j = 1, \ldots, n\}$, A_j is the jth column of a A, and $\|A_j\|_1$ is

* Address after July 1, 1998: Department of Mathematics, Box 5054, Tennessee Technological University, Cookeville, TN 38505, E-mail: rablamowicz@tntech.edu

the 1-vector norm on $\mathbb{C}^n$ defined as $\|\mathbf{x}\|_1 = \sum_{i=1}^{n} |x_i|$. However, for several reasons, there is no obvious way[1] to implement definition (1) on a computer, unless of course A is diagonalizable, that is, when A has a complete set of linearly independent eigenvectors (cf. [2]).

Another approach to solving $\mathbf{x}' = A\mathbf{x}$ is to find Jordan canonical form J of the matrix A. Let P be a nonsingular matrix such that $P^{-1}AP = J$. Then, if a change of basis is made such that $\mathbf{x} = P\mathbf{y}$, the matrix equation $\mathbf{x}' = A\mathbf{x}$ is transformed into $\mathbf{y}' = J\mathbf{y}$ and, at least theoretically, its solution is represented as $e^{Jt}\mathbf{c}$ for some constant vector $\mathbf{c}$. However, since the Jordan form is extremely discontinuous on a set of all $n \times n$ matrices, numeric computations of J are seriously ill-posed (cf. [2, 3]).

In this paper we present another approach to exponentiate a matrix, let it be numeric or symbolic, with real, complex, or quaternionic entries, totally different from the linear algebra methods. It relies on the well-known isomorphism between matrix algebras over $\mathbb{R}$, $\mathbb{C}$, or $\mathbb{H}$, and simple orthogonal Clifford algebras (cf. [4, 5, 6, 7]). This is not a matrix method in the sense that elements of the real Clifford algebra $C\ell(Q)$ are not viewed here as matrices but instead they are treated as symbolic multivariate polynomials in some basis Grassmann monomials. This is possible due to the linear isomorphism $C\ell(V, Q) \simeq \bigwedge V$. The critical exponentiation is done in the real Clifford algebra $C\ell_{p,q}$ over Q with a suitable signature (p, q) depending whether the given matrix A has real, complex, or quaternionic entries. Three examples of computation of the matrix exponential with a Maple package 'CLIFFORD' (cf. [8, 9, 10]) are presented below. The Reader is encouraged to repeat these computations.

In order to find matrix exponential e^A, the following steps will be taken:

- We will view elements of $C\ell_{p,q}$ as real multivariate polynomials in basis Grassmann or Clifford monomials.
- We will find explicit spinor (left-regular) representation γ of $C\ell_{p,q}$ in a minimal left ideal $S = C\ell_{p,q}f$ generated by a primitive idempotent f.
- For a matrix A (numeric or symbolic) in the matrix ring $\mathbb{R}(n)$, $\mathbb{C}(n)$ or $\mathbb{H}(n)$ where $n = 2^{m-1}$, $m = \left[\frac{1}{2}(p+q)\right]$, we will find its isomorphic image $p = \varphi(A)$ in $C\ell_{p,q}$.[2]
- We will find a *real* minimal polynomial $p(x)$ of p and then a formal power series $\exp(p) \bmod p(x)$ in $C\ell_{p,q}$.
- We will check the truncation error of the power series $\exp(p)$ in $C\ell_{p,q}$ via a polynomial norm, or in a matrix norm, both built into Maple.[3]
- We will map $\exp(p)$ back to the matrix ring $\mathbb{R}(n)$, $\mathbb{C}(n)$ or $\mathbb{H}(n)$ to get $\exp(A)$.

Before we proceed, let's recall certain useful facts about orthogonal Clifford algebras $C\ell_{p,q}$. For more information see [4].

[1] It is possible to compute the exponential e^{At} with a help of the Laplace transform method applied to an appropriate system of differential equations [1].

[2] The brackets $[\cdot]$ denote the floor function

[3] It is also possible to use the $\mathbb{R}^{n'}$ topology where $n' = 2^n$, $n = p + q$.

- If $p - q \neq 1 \bmod 4$ then $C\ell_{p,q}$ is a simple algebra of dimension 2^n, $n = p + q$, isomorphic with a full matrix algebra with entries in $\mathbb{R}$, $\mathbb{C}$, or $\mathbb{H}$.
- If $p - q = 1 \bmod 4$ then $C\ell_{p,q}$ is a semi-simple algebra of dimension 2^n, $n = p + q$, containing two copies of a full matrix algebra with entries in $\mathbb{R}$ or $\mathbb{H}$ projected out by two central idempotents $\frac{1}{2}(1 \pm \mathbf{e}_1\mathbf{e}_2 \cdots \mathbf{e}_n)$.[4]
- $C\ell_{p,q}$ has a faithful representation as a matrix algebra with entries in $\mathbb{R}$, $\mathbb{C}$, $\mathbb{H}$ or $\mathbb{R} \oplus \mathbb{R}$, $\mathbb{H} \oplus \mathbb{H}$ depending whether $C\ell_{p,q}$ is simple or semisimple.
- Any primitive idempotent f in $C\ell_{p,q}$ is expressible as a product

$$f = \frac{1}{2}(1 \pm e_{T_1})\frac{1}{2}(1 \pm e_{T_2}) \cdots \frac{1}{2}(1 \pm e_{T_k}) \tag{2}$$

where $\{e_{T_1}, e_{T_2}, \ldots, e_{T_k}\}$, $k = q - r_{q-p}$, is a set of commuting basis monomials with square 1, and r_i is the Radon-Hurwitz number defined by the recursion $r_{i+8} = r_i + 4$ and

i	0	1	2	3	4	5	6	7
r_i	0	1	2	2	3	3	3	3

.

- $C\ell_{p,q}$ has a complete set of 2^k primitive idempotents each with k factors as in (2).
- The division ring $\mathbb{K} = fC\ell_{p,q}f$ is isomorphic to $\mathbb{R}$ or $\mathbb{C}$ or $\mathbb{H}$ when $(p - q) \bmod 8$ is $0, 1, 2$, or $3, 7$ or $4, 5, 6$.
- The mapping $S \times \mathbb{K} \to S$, or $(\psi, \lambda) \to \psi\lambda$ defines a right $\mathbb{K}$-linear structure on the spinor space $S = C\ell_{p,q}f$ (cf. [7]).

Example 1. In $C\ell_{3,1} \simeq \mathbb{R}(4)$ we have $k = 2$ and $f = \frac{1}{2}(1 + \mathbf{e}_1)\frac{1}{2}(1 + \mathbf{e}_{34})$, $\mathbf{e}_{34} = \mathbf{e}_3\mathbf{e}_4 = \mathbf{e}_3 \wedge \mathbf{e}_4$ is a primitive idempotent. The ring $\mathbb{K} \simeq \mathbb{R}$ is just spanned by $\{1\}_{\mathbb{R}}$ and a real basis for $S = C\ell_{3,1}f$ may be generated by $\{1, \mathbf{e}_2, \mathbf{e}_3, \mathbf{e}_{23}\}_{\mathbb{R}}$ (here $\mathbf{e}_{23} = \mathbf{e}_2\mathbf{e}_3 = \mathbf{e}_2 \wedge \mathbf{e}_3$.)

Example 2. In $C\ell_{3,0} \simeq \mathbb{C}(2)$ we have $k = 1$ and $f = \frac{1}{2}(1 + \mathbf{e}_1)$ is a primitive idempotent. The ring $\mathbb{K} \simeq \mathbb{C}$ may be spanned by $\{1, \mathbf{e}_{23}\}_{\mathbb{R}}$ and a basis for $S = C\ell_{3,0}f$ over $\mathbb{K}$ may be generated by $\{1, \mathbf{e}_2\}_{\mathbb{K}}$.

Example 3. In $C\ell_{1,3} \simeq \mathbb{H}(2)$, the Clifford polynomial $f = \frac{1}{2}(1+\mathbf{e}_{14})$, $\mathbf{e}_{14} = \mathbf{e}_1\mathbf{e}_4 = \mathbf{e}_1 \wedge \mathbf{e}_4$, is a primitive idempotent. Thus, the ring $\mathbb{K} \simeq \mathbb{H}$ may be spanned by $\{1, \mathbf{e}_2, \mathbf{e}_3, \mathbf{e}_{23}\}_{\mathbb{R}}$ and a basis for $S = C\ell_{1,3}f$ as a right-quaternionic space over $\mathbb{K}$ may be generated by $\{1, \mathbf{e}_1\}_{\mathbb{K}}$.

2 Exponential of a real matrix

We now proceed to exponentiate a real 4×4 matrix using the spinor representation γ of $C\ell_{3,1}$ from Example 1. Instead of $C\ell_{3,1}$ one could also use $C\ell_{2,2}$, the Clifford algebra of the neutral signature $(2, 2)$, since $C\ell_{2,2} \simeq \mathbb{R}(4)$. From now on $\mathbf{e}_{ij} = \mathbf{e}_i\mathbf{e}_j = \mathbf{e}_i \wedge \mathbf{e}_j$, $i \neq j$, $\mathbb{K} = \{Id\}_{\mathbb{R}} \simeq \mathbb{R}$, and Id denotes the unit element of $C\ell_{3,1}$ in 'CLIFFORD'

Recall the following facts about the simple algebra $C\ell_{3,1} \simeq \mathbb{R}(4)$ and its spinor space S:

[4] For the purpose of this paper, it is enough to consider simple Clifford algebras only.

- $C\ell_{3,1} = \{1, \mathbf{e}_i, \mathbf{e}_{ij}, \mathbf{e}_{ijk}, \mathbf{e}_{ijkl}\}_{\mathbb{R}}, \quad i<j<k<l,\ i,j,k,l = 1,\dots,4.$
- $S = C\ell_{3,1}f = \{f_1 = f,\ f_2 = \mathbf{e}_2 f,\ f_3 = \mathbf{e}_3 f,\ f_4 = \mathbf{e}_{23} f\}_{\mathbb{K}}.$
- Each basis monomial $\mathbf{e}_{ijkl}$ has a unique matrix $\gamma_{\mathbf{e}_{ijkl}}$ representation in the spinor basis f_i, $i = 1,\dots,4$. For example, the basis 1-vectors $\mathbf{e}_1, \mathbf{e}_2, \mathbf{e}_3, \mathbf{e}_4$ are represented under γ as:

$$\begin{aligned}\gamma_{\mathbf{e}_1} &= \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \quad \gamma_{\mathbf{e}_2} = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix} \\ \gamma_{\mathbf{e}_3} &= \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \end{pmatrix}, \quad \gamma_{\mathbf{e}_4} = \begin{pmatrix} 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \end{pmatrix}.\end{aligned} \tag{3}$$

Since $\gamma : \mathbb{R}(4) \to C\ell_{3,1}$ is a linear isomorphism of algebras, matrices representing Clifford monomials of higher ranks are matrix products of matrices shown in (3). For example, $\gamma_{\mathbf{e}_{ijkl}} = \gamma_{\mathbf{e}_i}\gamma_{\mathbf{e}_j}\gamma_{\mathbf{e}_k}\gamma_{\mathbf{e}_l}$:

$$\gamma_{\mathbf{e}_{1234}} = \gamma_{\mathbf{e}_1}\gamma_{\mathbf{e}_2}\gamma_{\mathbf{e}_3}\gamma_{\mathbf{e}_4} = \begin{pmatrix} 0 & 1 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & -1 & 0 \end{pmatrix}. \tag{4}$$

Then, a matrix representing any Clifford polynomial may be found by the linearity of γ.

Relevant information about $C\ell_{3,1}$ is stored in 'CLIFFORD' and can be retrieved as follows:

```
> restart:with(Cliff3):dim:=4:B:=linalg[diag](1,1,1,-1):
> eval(makealiases(dim)):data:=clidata();
```

$$\begin{aligned} &data := \\ &\quad [real,\ 4,\ simple,\ cmulQ(\frac{1}{2}\,Id + \frac{1}{2}\,e1,\ \frac{1}{2}\,Id + \frac{1}{2}\,e34), \\ &\quad [Id,\ e2,\ e3,\ e23], [Id], [Id,\ e2,\ e3,\ e23]] \end{aligned}$$

In the Maple list *data* above,

- *real,* 4, and *simple* mean that $C\ell_{3,1}$ is a simple algebra isomorphic to $\mathbb{R}(4)$.
- The fourth element `data[4]` in the list `'data'` is a primitive idempotent f written as a Clifford product of two Clifford polynomials (Clifford product in orthogonal Clifford algebras is realized in 'CLIFFORD' through a procedure `'cmulQ'`).
- The list $[Id,\ e2,\ e3,\ e23]$ contains generators of the spinor space $S = C\ell_{3,1}f$ over the reals $\mathbb{R}$ (compare with Example 1 above).

– The list $[Id]$ contains the only basis element of the field $\mathbb{K} \subset C\ell_{3,1}$, that is, the identity element of $C\ell_{3,1}$.

– The final list $[Id,\, e2,\, e3,\, e23]$ contains generators of the spinor space $S = C\ell_{3,1}f$ over the field $\mathbb{K}$. In this case it coincides with `data[5]` since $\mathbb{K} \simeq \mathbb{R}$.

Thus, a real spinor basis in S consists of the following four polynomials:

```
> f1:=f;f2:=cmulQ(e2,f);f3:=cmulQ(e3,f);f4:= cmulQ(e23,f);
```

$$\begin{aligned} f1 &:= \frac{1}{4}\,Id + \frac{1}{4}\,e34 + \frac{1}{4}\,e1 + \frac{1}{4}\,e134, & f2 &:= \frac{1}{4}\,e2 + \frac{1}{4}\,e234 - \frac{1}{4}\,e12 - \frac{1}{4}\,e1234 \\ f3 &:= \frac{1}{4}\,e3 + \frac{1}{4}\,e4 - \frac{1}{4}\,e13 - \frac{1}{4}\,e14, & f4 &:= \frac{1}{4}\,e23 + \frac{1}{4}\,e24 + \frac{1}{4}\,e123 + \frac{1}{4}\,e124 \end{aligned} \tag{5}$$

Procedure 'matKrepr' allows us now to compute 16 matrices $m[i]$ representing each basis monomial in $C\ell_{3,1}$.

```
> for i from 1 to 16 do
> lprint (`The basis element`,clibas[i],
          `is represented by the following matrix:);
> m[i]:=subs(Id=1,matKrepr(clibas[i])) od:
```

Let's define a 4×4 real matrix A without a complete set of eigenvectors. Therefore, A cannot be diagonalized.

```
> A:=linalg[matrix](4,4,[0,1,0,0,-1,2,0,0,-1,1,1,0,-1,1,0,1]);
> linalg[eigenvects](A);#A has incomplete set of eigenvectors
```

$$A := \begin{bmatrix} 0 & 1 & 0 & 0 \\ -1 & 2 & 0 & 0 \\ -1 & 1 & 1 & 0 \\ -1 & 1 & 0 & 1 \end{bmatrix}$$

$$[1,\, 4,\, \{[0,\, 0,\, 1,\, 0],\, [1,\, 1,\, 0,\, 0],\, [0,\, 0,\, 0,\, 1]\}] \tag{6}$$

Maple output in (6) shows that A has only one eigenvalue $\lambda = 1$ with an algebraic multiplicity 4 and a geometric multiplicity 3.

In the Appendix, one can find a procedure 'phi' which gives the isomorphism φ from $\mathbb{R}(4)$ to $C\ell_{3,1}$. It can find the image $p = \mathtt{phi}(A)$ of any real 4×4 matrix A using the previously computed matrices $m[i]$. In particular, the image p of A under φ is computed as follows:

```
> FBgens:=[Id]; #assigning a basis element of K
> p:=phi(A,m,FBgens); #finding the image of A in Cl(3,1)
```

$$p := Id - \frac{1}{2}\,e1 - \frac{1}{2}\,e3 - \frac{1}{2}\,e4 + \frac{1}{2}\,e12 - \frac{1}{2}\,e23 - \frac{1}{2}\,e24 - \frac{1}{2}\,e134 + \frac{1}{2}\,e1234 \tag{7}$$

Let's go back to the exponentiation problem. So far we have found a Clifford polynomial p in $C\ell_{3,1}$ which is the isomorphic image of A. We will now compute a sequence of finite power series expansions of p up to a specified order N. Procedure `'sexp'` (defined in the Appendix) finds these expansions, which are just Clifford polynomials, modulo the minimal polynomial $p(x)$ of p. The minimal polynomial $p(x)$ can be computed using a procedure `'climinpoly'`.

```
> p(x)=climinpoly(p);
```

$$p(x) = x^2 - 2\,x + 1 \tag{8}$$

It can be easily verified that the polynomial (8) is satisfied by $p = \varphi(A)$ and that it is also the minimal polynomial of A.

```
> cmul(p,p)-2*p+Id; #p satisfies its own minimal polynomial
```

$$0$$

```
> linalg[minpoly](A,x); #matrix A has the same minimal polynomial as p
```

$$x^2 - 2\,x + 1$$

A finite sequence of say 20 Clifford polynomials approximating $\exp(p)$ can now be computed.

```
> N:=20:for i from 1 to N do p.i:=sexp(p,i) od:# we want 20 polynomials
```

For example, Maple displays polynomial p_{20} as follows:

```
> p_lim:=p.20;
```

$$\begin{aligned} p\_lim := \frac{6613313319248080001}{2432902008176640000}\,Id - \frac{82666416490601}{60822550204416}\,e1 - \frac{82666416490601}{60822550204416}\,e3 \\ - \frac{82666416490601}{60822550204416}\,e4 + \frac{82666416490601}{60822550204416}\,e12 - \frac{82666416490601}{60822550204416}\,e23 \\ - \frac{82666416490601}{60822550204416}\,e24 - \frac{82666416490601}{60822550204416}\,e134 + \frac{82666416490601}{60822550204416}\,e1234 \end{aligned}$$

Having computed the approximation polynomials $p_1, p_2, \ldots, p_N$, $N = 20$, one can show that the sequence converges to some limiting polynomial p_{lim} by verifying that $|p_i - p_j| < \epsilon$ for $i, j > M$, M sufficiently large, in one of the Maple's built-in polynomial norms.

Finally, we map back p_{lim} into a 4×4 matrix which approximates $\exp(A)$ up to and including the terms of order N.

```
> expA:=0:for i from 1 to nops(clibas) do
>     expA:=evalm(expA+coeff(p_lim, clibas[i])*m[i])od:
> evalm(expA); #the matrix exponent of A
```

$$\begin{bmatrix} \frac{1}{2432902008176640000} & \frac{82666416490601}{30411275102208} & 0 & 0 \\ \frac{-82666416490601}{30411275102208} & \frac{7775794614048301}{1430277488640000} & 0 & 0 \\ \frac{-82666416490601}{30411275102208} & \frac{82666416490601}{30411275102208} & \frac{6613313319248080001}{2432902008176640000} & 0 \\ \frac{-82666416490601}{30411275102208} & \frac{82666416490601}{30411275102208} & 0 & \frac{6613313319248080001}{2432902008176640000} \end{bmatrix}$$

Although A had an incomplete set of eigenvectors, Maple can find $\exp(A)$ in a closed form.

```
> mA:=linalg[exponential](A);
```

$$mA := \begin{bmatrix} 0 & e & 0 & 0 \\ -e & 2\,e & 0 & 0 \\ -e & e & e & 0 \\ -e & e & 0 & e \end{bmatrix}$$

Notice that our result is very close to the Maple closed-form result:

```
> map(evalf,evalm(expA));
```

$$[.411031762331216485 85\,10^{-18}\,,\ 2.7182818284590452349\,,\ 0\,,\ 0]$$
$$[-2.7182818284590452349\,,\ 5.4365636569180904703\,,\ 0\,,\ 0]$$
$$[-2.7182818284590452349\,,\ 2.7182818284590452349\,,\ 2.7182818284590452353\,,\ 0]$$
$$[-2.7182818284590452349\,,\ 2.7182818284590452349\,,\ 0\,,\ 2.7182818284590452353]$$

The 1-norm of the difference matrix between mA and $expA$ can be computed in Maple as follows:

```
> evalf(linalg[norm](mA-expA,1));
```

$$.2\,10^{-17}$$

3 Exponential of a complex matrix

In this section we exponentiate a complex 2×2 matrix using a spinor representation of $C\ell_{3,0} \simeq \mathbb{C}(2)$ (see Example 2 above). Note that instead of using $C\ell_{3,0}$, one could also use $C\ell_{1,2}$ since $C\ell_{1,2} \simeq \mathbb{C}(2)$. As before, $\mathbf{e}_{ijk} = \mathbf{e}_i\mathbf{e}_j\mathbf{e}_k = \mathbf{e}_i \wedge \mathbf{e}_j \wedge \mathbf{e}_k$, $i, j, k = 1, \ldots, 3$, $\mathbb{K} = \{Id, \mathbf{e}_{23}\}_{\mathbb{R}} \simeq \mathbb{C}$, $\mathbf{e}_{23}^2 = -Id$, where Id denotes the unit element of $C\ell_{3,0}$ in 'CLIFFORD'.

Recall these facts about the simple algebra $C\ell_{3,0}$ and its spinor space S:

- $C\ell_{3,0} = \{1, \mathbf{e}_i, \mathbf{e}_{ij}, \mathbf{e}_{ijk}\}_{\mathbb{R}},\ i < j < k$.
- $S = C\ell_{3,0}f = \{f_1 = f,\ f_2 = \mathbf{e}_2 f,\ f_3 = \mathbf{e}_3 f,\ f_4 = \mathbf{e}_{23} f\}_{\mathbb{R}}$.

– $S = C\ell_{3,0}f = \{f_1 = f,\ f_2 = \mathbf{e}_2 f\}_{\mathbb{K}}$.

For example, the basis 1-vectors are represented in the spinor basis $\{f_1, f_2\}$ by these three matrices in $\mathbb{K}(2)$ well known as the *Pauli matrices*:

$$\gamma_{\mathbf{e}_1} = \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \qquad \gamma_{\mathbf{e}_2} = \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, \qquad \gamma_{\mathbf{e}_3} = \begin{pmatrix} 0 & -\mathbf{e}_{23} \\ \mathbf{e}_{23} & 0 \end{pmatrix}. \tag{9}$$

The following information about $C\ell_{3,0}$ is stored in 'CLIFFORD':

```
> dim:=3:B:=linalg[diag](1,1,1):
> data:=clidata();
```

$$data := [complex,\ 2,\ simple,\ \frac{1}{2}\,Id + \frac{1}{2}\,e1,\ [Id,\ e2,\ e3,\ e23],\ [Id,\ e23],\ [Id,\ e2]]$$

Now we define a Grassmann basis in $C\ell_{3,0}$, assign a primitive idempotent to f, and generate a spinor basis for $S = C\ell_{3,0}f$.

```
> clibas:=cbasis(dim); #ordered basis in Cl(3,0)
```

$$clibas := [Id,\ e1,\ e2,\ e3,\ e12,\ e13,\ e23,\ e123]$$

```
> f:=data[4]; #a primitive idempotent in Cl(3,0)
```

$$f := \frac{1}{2}\,Id + \frac{1}{2}\,e1$$

```
> sbasis:=minimalideal(clibas,f,'left'); #find a real basis in Cl(B)f
```

$$sbasis := [[\frac{1}{2}\,Id + \frac{1}{2}\,e1,\ \frac{1}{2}\,e2 - \frac{1}{2}\,e12,\ \frac{1}{2}\,e3 - \frac{1}{2}\,e13,\ \frac{1}{2}\,e23 + \frac{1}{2}\,e123],\ [Id,\ e2,\ e3,\ e23],\ left]$$

```
> fbasis:=Kfield(sbasis,f); #find a basis for the field K
```

$$fbasis := [[\frac{1}{2}\,Id + \frac{1}{2}\,e1,\ \frac{1}{2}\,e23 + \frac{1}{2}\,e123],\ [Id,\ e23]]$$

```
> SBgens:=sbasis[2];#generators for a real basis in S
```

$$SBgens := [Id,\ e2,\ e3,\ e23]$$

```
> FBgens:=fbasis[2]; #generators for K
```

$$FBgens := [Id,\ e23]$$

In the above, `'sbasis'` is a $\mathbb{K}$-basis returned for $S = C\ell_{3,0}f$. Since in the current signature $(3,0)$ we have $\mathbb{K} = \{Id, e23\}_{\mathbb{R}} \simeq \mathbb{C}$, `cmulQ`$(e23, e23) = -Id$, and $C\ell_{3,0} \simeq \mathbb{C}(2)$, the output from `'spinorKbasis'` shown below has two basis vectors and their generators modulo f:

```
> Kbasis:=spinorKbasis(SBgens,f,FBgens,'left');
```

$$Kbasis := [[\frac{1}{2}\,Id + \frac{1}{2}\,e1,\ \frac{1}{2}\,e2 - \frac{1}{2}\,e12],\ [Id,\ e2],\ left]$$

```
> cmulQ(f,f); #verifying that f is an idempotent
```

$$\frac{1}{2}\,Id + \frac{1}{2}\,e1$$

Note that the second list in `'Kbasis'` contains generators of the first list modulo the idempotent f. Thus, the spinor basis in S over $\mathbb{K}$ consists of the following two polynomials:

```
> for i from 1 to nops(Kbasis[1]) do f.i:=Kbasis[1][i] od;
```

$$f1 := \frac{1}{2}\,Id + \frac{1}{2}\,e1, \qquad f2 := \frac{1}{2}\,e2 - \frac{1}{2}\,e12 \tag{10}$$

We are in a position now to compute matrices $m[i]$ representing basis elements in $C\ell_{3,0}$. We will only display Clifford-algebra valued matrices representing the 1-vectors $\{\mathbf{e}_1, \mathbf{e}_2, \mathbf{e}_3\}$ and the unit pseudoscalar $\mathbf{e}_{123} = \mathbf{e}_1\mathbf{e}_2\mathbf{e}_3$.

```
> for i from 1 to nops(clibas) do
> lprint (`The basis element`,clibas[i],
         `is represented by the following matrix:`);
> m[i]:=subs(Id=1,matKrepr(clibas[i])) od:
```

```
The basis element   e1   is represented by the following matrix:
```

$$m_2 := \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$$

```
The basis element   e2   is represented by the following matrix:
```

$$m_3 := \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

```
The basis element   e3   is represented by the following matrix:
```

$$m_4 := \begin{bmatrix} 0 & -e23 \\ e23 & 0 \end{bmatrix}$$

```
The basis element   e123   is represented by the following matrix:
```

$$m_8 := \begin{bmatrix} e23 & 0 \\ 0 & e23 \end{bmatrix}$$

As an example, let's define a complex $2{\times}2$ matrix A and let's find its eigenvectors:

```
> A:=linalg[matrix](2,2,[1+2*I,1-3*I,1-I,-2*I]); #defining A
> linalg[eigenvects](A);
```

$$A := \begin{bmatrix} 1+2\,I & 1-3\,I \\ 1-I & -2\,I \end{bmatrix}$$

$$[\frac{1}{2}+\frac{1}{2}\sqrt{-23-8\,I},\,1,\,\{\left[-\frac{3}{4}+\frac{1}{4}\sqrt{-23-8\,I}+I+\frac{1}{2}\,I\,(\frac{1}{2}+\frac{1}{2}\sqrt{-23-8\,I}),\,1\right]\}],$$
$$[\frac{1}{2}-\frac{1}{2}\sqrt{-23-8\,I},\,1,\,\{\left[-\frac{3}{4}-\frac{1}{4}\sqrt{-23-8\,I}+I+\frac{1}{2}\,I\,(\frac{1}{2}-\frac{1}{2}\sqrt{-23-8\,I}),\,1\right]\}]$$

The image of A in $C\ell_{3,0}$ under the isomorphism $\varphi : \mathbb{C}(2) \rightarrow C\ell_{3,0}$ can now be computed. Recall that `'FBgens'` defined above contained the basis elements of the complex field $\mathbb{K}$ in $C\ell_{3,0}$.

```
> evalm(A);p:=phi(A,m,FBgens); #finding image of A in Cl(3,0)
```

$$\begin{bmatrix} 1+2\,I & 1-3\,I \\ 1-I & -2\,I \end{bmatrix}$$

$$p := \frac{1}{2}\,Id + \frac{1}{2}\,e1 + e2 + e3 + 2\,e13 + 2\,e23$$

Thus, we have found a Clifford polynomial p in $C\ell_{3,0}$ which is the isomorphic image of A. We will now compute a sequence of finite power expansions of p up to and including power $N = 30$ using the procedure `'sexp'`. This sequence of Clifford polynomials should converge to a polynomial p_{lim}, the image under φ of the matrix exponential $\exp(A)$. First, we find the *real* minimal polynomial $p(x)$ of p (called `'pol'` in Maple).

```
> pol:=climinpoly(p); #find the real minimal polynomial of p
```

$$pol := x^4 - 2\,x^3 + 13\,x^2 - 12\,x + 40$$

```
> &c(p$4)-2*&c(p$3)+13*&c(p$2)-12*p+40*Id;#checking that p satisfies pol
```

$$0$$

Observe that matrix A has the following *complex* minimal polynomial `'pol2'`:

```
> pol2:=linalg[minpoly](A,x);
```

$$pol2 := 6 + 2\,I - x + x^2$$

```
> evalm(&*(A$2)-A+6+2*I);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Furthermore, since $\{Id, e123\}_{\mathbb{R}}$ is another copy of the complex field $\mathbb{K}$ in $C\ell_{3,0}$, we can easily verify that the Clifford polynomial p also satisfies the complex minimal polynomial `'pol2'` of A if we replace 1 with Id and I with $e123$, namely:

```
> &c(p$2)-p+6*Id+2*e123;
```

$$0$$

On the other hand, matrix A of course satisfies the polynomial `'pol'`:

```
> evalm(&*(A$4)-2*&*(A$3)+13*&*(A$2)-12*A+40);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

As expected, the complex minimal polynomial of A is a factor of the real minimal polynomial of p:

```
> divide(pol,pol2);
```

true

```
> pol3:=quo(pol,pol2,x);
```

$$pol3 := x^2 - x + 6 - 2\,I$$

Let's check that `pol3` $*$ `pol2` $=$ `pol`:

```
> pol;expand(pol3 * pol2);
```

$$x^4 - 2\,x^3 + 13\,x^2 - 12\,x + 40$$
$$x^4 - 2\,x^3 + 13\,x^2 - 12\,x + 40$$

The following loop computes Clifford polynomials p_i approximating $\exp(p)$ in $C\ell_{3,0}$. We will only display polynomial p_{30} and assign it to p_{lim}.

```
> Digits:=20:
> N:=30:for i from 1 to N do p.i:=sexp(p,i) od;
> p_lim:=p.N:
```

$$p30 := -\frac{73941882654520889827560020338 9}{544108430383981658741145600000}\,Id + \frac{14060661868676909855563160922593 9}{1768352398747940390908723200000000}\,e1$$
$$-\frac{13294860446171527820401106221093}{8841761993739701954543616000000 0}\,e2 + \frac{542937608544885918642044746589 3}{12631088562485288506490880000000}\,e3$$
$$+\frac{50830755859220399836279191881837}{44208809968698509772718080000000}\,e13 + \frac{15796535483801410769637551225479}{22104404984349254886359040000000}\,e23$$
$$-\frac{537129223345642211370021843709}{118416455273299579748352000000 00}\,e123 - \frac{2456920164957545120945605291 3}{84691206836587183472640000000}\,e12$$

By picking up numeric coefficients of the basis monomials in the subsequent approximations to $\exp(p)$, one can get an idea about the approximation errors.

```
> sort([op(L:=cliterms(p_lim))],bygrade):
> for i from 1 to nops(L) do
>     L.i:=map(evalf,[seq(coeff(p.j,L[i]),j= 1..N)]) od:
> approxerror:=
        max(seq(min(seq(abs(L.j[i]-L.j[i-1]), i=2..N)), j=1..nops(L)));
```

$$approxerror := .1\,10^{-19}$$

Having computed the finite sequence of polynomials p_i one can again show by using Maple's built-in polynomial norm functions that this is a convergent sequence. For example, in the infinity norm one gets $|p_{29} - p_{30}| < .6 \times 10^{-20}$ and $|p_i - p_j| \to 0$ as $i, j \to \infty$.

Thus, we have found an approximation p_{lim} to the power series expansion of $\exp(p)$ in $C\ell_{3,0}$ up to and including terms of degree $N = 30$. Finally, we map back p_{lim} into a 2×2 complex matrix which approximates $\exp(A)$. We expand p_{lim} over the matrices $m[i]$:

```
> expA:=0:  for i from 1 to nops(clibas) do
>     expA:=evalm(expA+coeff(p_lim,clibas[i] )*m[i]) od:
> evalm(expA); #the matrix exponent of A
```

$$\left[-\frac{71217499957445566702813183482 49}{12631088562485288506490880000000} + \frac{59690930841545753352384242857 7}{22866625845878539537612800000000}\,e23,\right.$$
$$\left.-\frac{77890213936656597766146450924 53}{1768352398747940390908723200000 0} - \frac{62281892671831801104794433 01}{39428147129274033242112000000}\,e23\right]$$
$$\left[\frac{12355386075985243242271013020079}{8841761993739701954543616000000 0} - \frac{34040577069678494848992113102 9}{47282149699142791200768000000 0}\,e23,\right.$$
$$\left.-\frac{31743144776163499207933472943947}{1473626999895661699242393600000 00} - \frac{59591417660584766265872213018 57}{51010165348498280506982400000 00}\,e23\right]$$

Maple can find the exponent of A in a closed form with its `'linalg[exponential]'` command. We won't display the result but we will just compare it numerically with our result saved in `'expA'`.

```
> mA:=linalg[exponential](A):
```

Let's replace the monomial $e2we3$ in `'expA'` with the imaginary unit I used by Maple and let's apply `'evalf'` to the entries of `'expA'`:

```
> fexpA:=subs(e2we3=I,map(evalf,evalm(expA)));
```

$$fexpA := \begin{matrix} [-.5638270969690108 5353 + .26103952215715461164\, I\,, \\ -.44046771442052942162 - 1.5796302186766888057\, I] \\ [.13973895796712599250 - .71994563035478140661\, I\,, \\ -2.1540827359052753813 - 1.1682263182928324795\, I] \end{matrix}$$

```
> fmA:=map(evalf,mA); #applying 'evalf' to mA
```

$$fmA := \begin{matrix} [-.56382709696901085362 + .26103952215715461158\, I\,, \\ -.44046771442052942180 - 1.5796302186766888058\, I] \\ [.13973895796712599243 - .71994563035478140663\, I\,, \\ -2.1540827359052753816 - 1.1682263182928324795\, I] \end{matrix}$$

Let's check the 1-norm of the difference matrix between `'fmA'` and `'fexpA'`:

```
> evalf(linalg[norm](fmA-fexpA,1));
```

$$.5059126028\, 10^{-18}$$

The floating-point approximation `'fexpA'` to $\exp(A)$ is within approximately $.5 \times 10^{-18}$ in the matrix $\|\cdot\|_1$ norm to the closed matrix exponential computed by Maple.

4 Exponential of a quaternionic matrix

In order to exponentiate a quaternionic 2×2 matrix, we will use the spinor representation of $C\ell_{1,3} \simeq \mathbb{H}(2)$ (see Example 3 above). Note that two other algebras could be used instead of $C\ell_{1,3}$, namely, $C\ell_{0,4}$ and $C\ell_{4,0}$ since both are isomorphic to $\mathbb{H}(2)$. As before $\mathbf{e}_{ij} = \mathbf{e}_i\mathbf{e}_j = \mathbf{e}_i \wedge \mathbf{e}_j$, $i, j = 1, \ldots, 4$, but this time $\mathbb{K} = \{Id, \mathbf{e}_2, \mathbf{e}_3, \mathbf{e}_{23}\}_{\mathbb{R}} \simeq \mathbb{H}$.

Recall the following facts about the simple algebra $C\ell_{1,3}$ and its spinor space S:

- $C\ell_{1,3} = \{1, \mathbf{e}_i, \mathbf{e}_{ij}, \mathbf{e}_{ijk}, \mathbf{e}_{ijkl}\}_{\mathbb{R}},\ i < j < k < l.$
- $S = C\ell_{1,3}f = \{f_1 = f, f_2 = \mathbf{e}_2 f, f_3 = \mathbf{e}_3 f, f_4 = \mathbf{e}_{23} f\}_{\mathbb{R}}.$
- $S = C\ell_{1,3}f = \{f_1 = f, f_2 = \mathbf{e}_1 f\}_{\mathbb{K}}.$

For example, the basis 1-vectors $\mathbf{e}_1, \mathbf{e}_2, \mathbf{e}_3, \mathbf{e}_4$ are represented by:

$$\gamma_{\mathbf{e}_1} = \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, \ \gamma_{\mathbf{e}_2} = \begin{pmatrix} \mathbf{e}_2 & 0 \\ 0 & -\mathbf{e}_2 \end{pmatrix}, \ \gamma_{\mathbf{e}_3} = \begin{pmatrix} \mathbf{e}_3 & 0 \\ 0 & -\mathbf{e}_3 \end{pmatrix}, \ \gamma_{\mathbf{e}_4} = \begin{pmatrix} 0 & -1 \\ 1 & 0 \end{pmatrix}. \tag{11}$$

In order to compute the spinor representation of $C\ell_{1,3}$, we proceed as follows:

```
> data:=clidata(linalg[diag](1,-1,-1,-1));
```

$$data := [quaternionic, 2, simple, \frac{1}{2}\,Id + \frac{1}{2}\,e14, [Id, e1, e2, e3, e12, e13, e23, e123], [Id, e2, e3, e23], [Id, e1]]$$

We define a Grassmann basis in $C\ell_{1,3}$, assign a primitive idempotent to f, and generate a spinor basis for $S = C\ell_{1,3}f$.

```
> clibas:=cbasis(dim); #ordered basis in Cl(1,3)
```

$$clibas := [Id, e1, e2, e3, e4, e12, e13, e14, e23, e24, e34, e123, e124, e134, e234, e1234]$$

```
> f:=data[4]; #a primitive idempotent in Cl(1,3)
```

$$f := \frac{1}{2}\,Id + \frac{1}{2}\,e14$$

Next, we compute a real basis in the spinor space $S = C\ell_{1,3}f$ using the command `'minimalideal'`:

```
> sbasis:=minimalideal(clibas,f,'left');#find a real basis in Cl(B)f
```

$$sbasis := [[\frac{1}{2}\,Id + \frac{1}{2}\,e14, \frac{1}{2}\,e1 + \frac{1}{2}\,e4, \frac{1}{2}\,e2 - \frac{1}{2}\,e124, \frac{1}{2}\,e3 - \frac{1}{2}\,e134, \frac{1}{2}\,e12 - \frac{1}{2}\,e24, \frac{1}{2}\,e13 - \frac{1}{2}\,e34, \frac{1}{2}\,e23 + \frac{1}{2}\,e1234, \frac{1}{2}\,e123 + \frac{1}{2}\,e234], [Id, e1, e2, e3, e12, e13, e23, e123], left]$$

In the following, we compute a basis for the subalgebra $\mathbb{K}$:

```
> fbasis:=Kfield(sbasis,f); #a basis for the field K
```

$$fbasis := [[\frac{1}{2}\,Id + \frac{1}{2}\,e14, \frac{1}{2}\,e2 - \frac{1}{2}\,e124, \frac{1}{2}\,e3 - \frac{1}{2}\,e134, \frac{1}{2}\,e23 + \frac{1}{2}\,e1234], [Id, e2, e3, e23]]$$

```
> SBgens:=sbasis[2];#generators for a real basis in S
```

$$SBgens := [Id, e1, e2, e3, e12, e13, e23, e123]$$

Thus, a possible set of generators for $\mathbb{K}$ is:

```
> FBgens:=fbasis[2]; #generators for K
```

$$FBgens := [Id,\, e2,\, e3,\, e23] \tag{12}$$

In the above, `'sbasis'` is a real basis for $S = C\ell_{1,3}f$. Since in the current signature $(1,3)$ we have that $\mathbb{K} = \{Id,\, e2,\, e3,\, e23\}_{\mathbb{R}} \simeq \mathbb{H}$ and $C\ell_{1,3} = \mathbb{H}(2)$, the output from `'spinorKbasis'` shown below has two basis vectors and their generators modulo f for S over $\mathbb{K}$:

```
> Kbasis:=spinorKbasis(SBgens,f,FBgens,'left');
```

$$Kbasis := [[\frac{1}{2}\,Id + \frac{1}{2}\,e14,\, \frac{1}{2}\,e1 + \frac{1}{2}\,e4],\, [Id,\, e1],\, left]$$

```
> cmulQ(f,f); #f is an idempotent in Cl(1,3)
```

$$\frac{1}{2}\,Id + \frac{1}{2}\,e14$$

Notice that the generators of the first list in `'Kbasis'` are listed in `Kbasis[2]`. Furthermore, a spinor basis in S over $\mathbb{K}$ consists of the following two polynomials f_1 and f_2:

```
> for i from 1 to nops(Kbasis[1]) do f.i:=Kbasis[1][i] od;
```

$$f1 := \frac{1}{2}\,Id + \frac{1}{2}\,e14, \qquad f2 := \frac{1}{2}\,e1 + \frac{1}{2}\,e4 \tag{13}$$

Using the procedure `'matKrepr'` we can now find matrices $m[i]$ with entries in $\mathbb{K}$ representing basis monomials in $C\ell_{1,3}$. Below we will display only matrices representing the 1-vectors $\mathbf{e}_1, \mathbf{e}_2, \mathbf{e}_3$ and $\mathbf{e}_4$:

```
> for i from 1 to nops(clibas) do
> lprint (`The basis element`,clibas[i],
          `is represented by the following matrix:`);
> m[i]:=subs(Id=1,matKrepr(clibas[i])) od;
```

```
The basis element   e1   is represented by the following matrix:
```

$$m_2 := \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

```
The basis element   e2   is represented by the following matrix:
```

$$m_3 := \begin{bmatrix} e2 & 0 \\ 0 & -e2 \end{bmatrix}$$

```
The basis element   e3   is represented by the following matrix:
```

$$m_4 := \begin{bmatrix} e3 & 0 \\ 0 & -e3 \end{bmatrix}$$

```
The basis element   e4   is represented by the following matrix:
```

$$m_5 := \begin{bmatrix} 0 & -1 \\ 1 & 0 \end{bmatrix}$$

Let's define a 2×2 quaternionic matrix A. In Maple, we will represent the standard quaternionic basis $\{1, \mathbf{i}, \mathbf{j}, \mathbf{k}\}$ as {`1`,`'ii'`,`'jj'`,`'kk'`}. Later we will make substitutions: `'ii'` $\rightarrow e2$, `'jj'` $\rightarrow e3$, `'kk'` $\rightarrow e2we3$ since, as we may recall from Example 3 above, $\mathbb{K} = \{1, \mathbf{e}_2, \mathbf{e}_3, \mathbf{e}_{23}\}_{\mathbb{R}}$.

```
> A:=linalg[matrix](2,2,[1+2*'ii'-3*'kk',2+'ii' -2*'jj',
> 'kk'-3*'ii',2*'kk'-2*'jj']); #defining a quaternionic matrix A
```

$$A := \begin{bmatrix} 1 + 2\,ii - 3\,kk & 2 + ii - 2\,jj \\ kk - 3\,ii & 2\,kk - 2\,jj \end{bmatrix} \tag{14}$$

The isomorphism $\varphi : \mathbb{H}(2) \rightarrow C\ell_{1,3}$ has been defined in Maple through the procedure `'phi'` (see the Appendix). This way we can find image p in $C\ell_{1,3}$ of any matrix A. Recall that `'FBgens'` in (12) contains the basis elements of the field $\mathbb{K}$.

```
> p:=phi(A,m,FBgens);#finding image of A in Cl(1,3)
```

$$p := \frac{1}{2}\,Id + e1 + e2 + e3 - e4 - 2\,e12 + e13 + \frac{1}{2}\,e14 - \frac{1}{2}\,e23 + e24 + e34 + \frac{1}{2}\,e123 - e124 + e134 + \frac{1}{2}\,e234 - \frac{5}{2}\,e1234$$

The minimal polynomial $p(x)$ of p in $C\ell_{1,3}$ is then found with the procedure `'climinpoly'`:

```
> climinpoly(p);
```

$$x^4 - 2\,x^3 + 16\,x^2 + 10\,x + 330$$

So far we have found a Clifford polynomial p in $C\ell_{1,3}$ which is the isomorphic image of the quaternionic matrix A. We will now compute a sequence of finite power expansions of p using the procedure `'sexp'`. This sequence of Clifford polynomials will be shown to converge to a polynomial p_{lim} that is the image of $\exp(A)$. For example, polynomial `p20 = sexp(p,20)` looks as follows:

```
> for i from 1 to 20 do p.i:=sexp(p,i) od;
```

$$
\begin{aligned}
p20 := & -\frac{68240889697169513}{10861169679360000}\,Id - \frac{50515123107772493}{9503523469440000}\,e34 \\
& + \frac{976049744897473}{638892334080000}\,e123 - \frac{76665127748453}{66691392768000}\,e234 \\
& + \frac{23336382714907219}{152056375511040000}\,e124 - \frac{1736342897976643}{1974758123520000}\,e134 \\
& + \frac{9030311044661089}{1407929402880000}\,e1234 + \frac{802551523836832291}{152056375511040000}\,e12 \\
& - \frac{907882088300711}{365520133440000}\,e13 + \frac{4304638284278411}{4472246338560000}\,e23 \\
& - \frac{360072975386539}{116162242560000}\,e24 - \frac{19812017405738017}{76028187755520000}\,e14 \\
& - \frac{1889118161676113}{703964701440000}\,e1 - \frac{277471312336316837}{152056375511040000}\,e2 \\
& - \frac{98120514192871531}{152056375511040000}\,e3 - \frac{25277099300039}{44722463385600}\,e4
\end{aligned}
$$

Thus, we have a finite sequence of Clifford polynomials p_i approximating $\exp(p)$. Next, for each of the 16 basis monomials present in all polynomials, we create a sequence s_j (or `sj` in Maple) of its coefficients.

```
> for j from 1 to nops(clibas) do
>     s.j:=map(evalf,[seq(coeff(p.i,clibas[j]),i=1..N)]) od:
```

For example, the sequence `s1` of the coefficients of the identity element Id is:

```
> s1;
```

$$
\begin{aligned}
[&1.500000000,\ -2.,\ -6.916666667,\ -18.66666667,\ -20.22500000,\ -10.85972222,\\
&-5.099206349,\ -3.980456349,\ -5.027722663,\ -6.129274691,\ -6.428549232,\\
&-6.368049418,\ -6.301487892,\ -6.280796253,\ -6.280315663,\ -6.282290205,\\
&-6.282986035,\ -6.283054064,\ -6.283026981,\ -6.283014787]
\end{aligned}
$$

Having computed the finite sequence of polynomials $p_1, p_2, \ldots, p_{20}$, one can again verify that this is a convergent sequence by using any of the Maple's built-in polynomial norm functions to estimate norms of the differences $p_i - p_j$ for $i, j = 1, \ldots, 20$. It can be again observed that $|p_i - p_j| \to 0$ as $i, j \to \infty$. Finally, we map back $p_{lim} \simeq p_{20}$ into a 2×2 matrix `'expA'` which approximates $\exp(A)$ up to and including terms of order $N = 20$. After expressing back the basis elements $\{Id, e2, e3, e2we3\}$ in terms of `{1, 'ii', 'jj', 'kk'}` we obtain:

```
> p_lim:=p20:
> expA:=0:for i from 1 to nops(clibas) do
>     expA:=evalm(expA+coeff(p_limit,clibas[i])*m[i]) od:
> sexpA:=subs({e2we3='kk',e3='jj',e2='ii'}, evalm(expA));
```

$$sexpA :=$$

$$\left[-\frac{58889470322671}{8999548740000}-\frac{301630543173}{152472320000}\,ii+\frac{1778894447566499}{7602818775552000}\,jj+\frac{5608156472444431793}{76028187755520000}\,kk,-\frac{10065855790684619}{4751761734720000}-\frac{5520266650930879}{2534272925184000}\,ii+\frac{748687448521121}{95995186560000}\,jj+\frac{203548165276035707}{76028187755520000}\,kk\right]$$

$$\left[-\frac{30874478783885813}{9503523469440000}+\frac{33523343384679259}{4001483566080000}\,ii+\frac{3844312687422001}{1357646209920000}\,jj+\frac{2613788546323897}{6911653432320000}\,kk,-\frac{228937105237224287}{38014093877760000}+\frac{127067464810704809}{76028187755520000}\,ii+\frac{38636486222845507}{25342729251840000}\,jj-\frac{414457945578965819}{76028187755520000}\,kk\right]$$

```
> fexpA:=map(evalf,evalm(sexpA)); #floating-point approximation
```

$$fexpA :=$$
$$[-6.543602577 - 1.978264272\,ii + .2339782783\,jj + 7.376417403\,kk\,,$$
$$-2.118341860 - 2.178244733\,ii + 7.799218642\,jj + 2.677272355\,kk]$$
$$[-3.248740205 + 8.377728618\,ii + 2.831601237\,jj + .3781712396\,kk\,,$$
$$-6.022426997 + 1.671320448\,ii + 1.524559010\,jj - 5.451372153\,kk]$$

Thus, matrix `'sexpA'` is the exponential of the quaternionic matrix A from (14) computed with the Clifford algebra $C\ell_{1,3}$.

5 Conclusions

We have translated the problem of matrix exponentiation e^A, $A \in \mathbb{K}(n)$, into the problem of computing e^p in the Clifford algebra $C\ell(Q)$ isomorphic to $\mathbb{K}(n)$. This approach, alternative to the standard linear algebra methods, is based on the spinor representation of $C\ell(Q)$. It should be equally applicable to other functions representable as power series. Another use for the isomorphism between $C\ell_{p,q}$ and appropriate matrix rings could be to finding the Jordan canonical form of A in terms of idempotent and nilpotent Clifford polynomials from $C\ell(Q)$ (see also [11] and [12] for more on the Jordan form and its relation to the Clifford algebra). Generally speaking, any linear algebra property of A can be related to a corresponding property of p, its isomorphic image in $C\ell(Q)$, and it can be stated in the purely symbolic non-matrix language of the Clifford algebra. These investigations are greatly facilitated with 'CLIFFORD'. At [9] interested Reader my find complete Maple worksheets with the above and other computations.

6 Acknowledgements

The author thanks Prof. Thomas McDonald, Department of Mathematics, Gannon University, Erie, PA, for a critical reading of this paper, and for bringing to the author's attention a way of finding the exponential e^{At} that involves solving a system of differential equations with the Laplace transform method.

7 Appendix

The procedures described in this Appendix will work provided the Maple package 'CLIFFORD' has been loaded first into a worksheet.[5] Procedure `'phi'` was used above to provide the isomorphism φ between the matrix algebras $\mathbb{R}(4)$, $\mathbb{C}(2)$, and $\mathbb{H}(2)$ and, respectively, the Clifford algebras $C\ell_{3,1}$, $C\ell_{3,0}$, and $C\ell_{1,3}$.

```
> phi:=proc(A::matrix,m::table,FBgens::list(climon))
  local N,n,cb,fb,AA,M,a,j,L,sys,vars,sol,p;global B;
  if nops(FBgens)=1 then AA:=evalm(A) elif
    nops(FBgens)=2 then fb:=op(remove(has,FBgens,Id));
    AA:=subs(I=fb,evalm(A)) elif
    nops(FBgens)=4 then fb:=sort(remove(has,FBgens,Id),bygrade);
    AA:=subs('ii'=fb[1],'jj'=fb[2],'kk'=fb[3],evalm(A))
    else ERROR(`wrong number of elements 'FBgens'`) fi;
  N:=nops([indices(m)]);n:=linalg[coldim](B):cb:=cbasis(n);
  M:=map(displayid,evalm(AA-add(a[j]*m[j],j=1..N)));
  L:=map(clicollect,convert(M,mlist));
  sys:=op(map(coeffs,L,FBgens));vars:=seq(a[j],j=1..N);
  sol:=solve(sys,vars); vars:=seq(a[j]*cb[j],j=1..N);
  p:=subs(sol,p);RETURN(p)
  end:
```

Procedure `'climinpoly'` finds a *real* minimal polynomial of any Clifford polynomial p in an arbitrary Clifford algebra $C\ell_{p,q}$.

```
> climinpoly:=proc(p::clipolynom,s::string)
  local dp,L,flag,pp,expr,a,k,eq,sys,vars,sol,poly;
  option remember;
  dp:=displayid(p):L:=[Id,dp];flag:=false:
  while not flag do
    pp:=cmul(L[nops(L)],dp):
    expr:=expand(add(a[k]*L[k],k=1..nops(L)));
    eq:=clicollect(pp-expr); sys:=coeffs(eq,cliterms(eq));
    vars:=seq(a[k],k=1..nops(L)); sol:=solve(sys,vars):
    if sol<> then flag:=true else L:=[op(L),pp] fi;
    od;
  poly:='x'^nops(L)-add(a[k]*'x'^(k-1),k=1..nops(L));
  if nargs=1 then RETURN(sort(subs(op(sol),poly)))
    else RETURN([sort(subs(op(sol),poly)),L]) fi;
  end:
```

Procedure `'sexp'` finds a finite formal power series expansion $\sum_{k=0}^{n}(p^k/k!)$ of any Clifford polynomial p up to and including the degree specified as its second argument. Computation of the powers of p in $C\ell_{p,q}$ is performed modulo the real minimal polynomial of p.

```
> sexp:=proc(p::clipolynom,n::posint) local i,d,L,Lp,pol,poly,k;
  pol:=climinpoly(p,'s');readlib(powmod);
  poly:=add(powmod('x',k,pol[1],'x')/k!,k=0..n);
  L:=[op(poly)];Lp:=[]:
  for i from 1 to nops(L) do
```

[5] To download 'CLIFFORD', see the Web site in [9].

```
    d:=degree(L[i]);
    if d=0 then Lp:=[op(Lp),L[i]*Id] else
      Lp:=[op(Lp),coeffs(L[i])*pol[2][d+1]] fi od;
  RETURN(add(Lp[i],i=1..nops(Lp)))
  end:
```

References

[1] McDonald Th., Private communication, 1998.

[2] Scheick J.T., Linear Algebra with Applications, McGraw-Hill Companies, Inc., New York, 1997, 398–401.

[3] Kwak J.H. and Hong S., Linear Algebra, Birkhäuser, Boston, 1997.

[4] Crumeyrolle A., Orthogonal and Symplectic Clifford Algebras: Spinor Structures, Kluwer, Dordrecht, 1990.

[5] Abłamowicz R., Lounesto P. (eds.), Clifford Algebras and Spinor Structures, A Special Volume Dedicated to the Memory of Albert Crumeyrolle (1919–1992), Kluwer, Dordrecht, 1995.

[6] Abłamowicz R., Lounesto P. and Parra J.M. (eds.), Clifford Algebras with Numeric and Symbolic Computations, Birkhäuser, Boston, 1996.

[7] Lounesto P., Scalar products of spinors and an extension of Brauer-Wall groups, *Found. Physics*, 1981, V.11, N 9/10, 721–740.

[8] Abłamowicz R., Clifford algebra computations with Maple, in: Geometric (Clifford) Algebras in Physics, ed. W. Baylis, Birkhäuser, Boston, 1996.

[9] Abłamowicz R., 'CLIFFORD' – Maple V Package for Clifford Algebra Computations, ver. 3, 1997, available at: `http://math.gannon.edu/rafal/cliff3`.

[10] Maple V Release 4 for Windows: Waterloo Maple Software, Waterloo, Ontario, 1996.

[11] Sobczyk G., The generalized spectral decomposition of a linear operator, *The College Mathematics Journal*, 1997, V.28, N 1, 27–38.

[12] Sobczyk G., Spectral integral domains in the classroom, *Aportaciones Matemáticas, Serie Comunicaciones*, 1997, V.20, 169–188.